Description

*biplot* displays a two-dimensional biplot of a dataset. A biplot simultaneously displays the observations (rows) and the relative positions of the variables (columns). Marker symbols (points) are displayed for observations, and arrows are displayed for variables. Observations are projected to two dimensions such that the distance between the observations is approximately preserved. The cosine of the angle between arrows approximates the correlation between the variables.

Quick start

Biplot of *v1*, *v2*, *v3*, and *v4*

```stata
biplot v1 v2 v3 v4
```

As above, but use standardized instead of centered variables

```stata
biplot v1 v2 v3 v4, std
```

As above, but graph dimension 3 versus 1 instead of 2 versus 1

```stata
biplot v1 v2 v3 v4, std dim(3 1)
```

Biplot with separate row and column plots

```stata
biplot v1 v2 v3 v4, separate
```

Only graph the column plot

```stata
biplot v1 v2 v3 v4, norow
```

Biplot with different color row markers for each category of *catvar*

```stata
biplot v1 v2 v3 v4, rowover(catvar)
```

Label row observations using the values in variable *mylabel*

```stata
biplot v1 v2 v3 v4, rowlabel(mylabell)
```

Store biplot coordinates in new variables *x* and *y*

```stata
biplot v1 v2 v3 v4, generate(x y)
```

Menu

Statistics > Multivariate analysis > Biplot
## Syntax

```
biplot varlist [if] [in] [, options]
```

<table>
<thead>
<tr>
<th>options</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Main</strong></td>
<td></td>
</tr>
<tr>
<td>rowover(varlist)</td>
<td>identify observations from different groups of <code>varlist</code>; may not be combined with <code>separate</code> or <code>norow</code></td>
</tr>
<tr>
<td>dim(# #)</td>
<td>two dimensions to be displayed; default is <code>dim(2 1)</code></td>
</tr>
<tr>
<td>std</td>
<td>use standardized instead of centered variables</td>
</tr>
<tr>
<td>alpha(#)</td>
<td>row weight = #; column weight = 1 – #; default is 0.5</td>
</tr>
<tr>
<td>stretch(#)</td>
<td>stretch the column (variable) arrows</td>
</tr>
<tr>
<td>mahalanobis</td>
<td>approximate Mahalanobis distance; implies <code>alpha(0)</code></td>
</tr>
<tr>
<td>xnegate</td>
<td>negate the data relative to the x axis</td>
</tr>
<tr>
<td>ynegate</td>
<td>negate the data relative to the y axis</td>
</tr>
<tr>
<td>autoaspect</td>
<td>adjust aspect ratio on the basis of the data; default aspect ratio is 1</td>
</tr>
<tr>
<td>separate</td>
<td>produce separate plots for rows and columns; may not be combined with <code>rowover()</code></td>
</tr>
<tr>
<td>nograph</td>
<td>suppress graph</td>
</tr>
<tr>
<td>table</td>
<td>display table showing biplot coordinates</td>
</tr>
<tr>
<td><strong>Rows</strong></td>
<td></td>
</tr>
<tr>
<td>rowopts(row_options)</td>
<td>affect rendition of rows (observations)</td>
</tr>
<tr>
<td>row#opts(row_options)</td>
<td>affect rendition of rows (observations) in the # group of <code>varlist</code> defined in <code>rowover()</code>; available only with <code>rowover()</code></td>
</tr>
<tr>
<td>rowlabel(varname)</td>
<td>specify label variable for rows (observations)</td>
</tr>
<tr>
<td>norow</td>
<td>suppress row points; may not be combined with <code>rowover()</code></td>
</tr>
<tr>
<td>generate(newvar_x newvar_y)</td>
<td>store biplot coordinates for observations in variables <code>newvar_x</code> and <code>newvar_y</code></td>
</tr>
<tr>
<td><strong>Columns</strong></td>
<td></td>
</tr>
<tr>
<td>colopts(col_options)</td>
<td>affect rendition of columns (variables)</td>
</tr>
<tr>
<td>negcol</td>
<td>include negative column (variable) arrows</td>
</tr>
<tr>
<td>negcolopts(col_options)</td>
<td>affect rendition of negative columns (variables)</td>
</tr>
<tr>
<td>nocolumn</td>
<td>suppress column arrows</td>
</tr>
</tbody>
</table>

Y axis, X axis, Titles, Legend, Overall

`twoway_options` any options other than by() documented in [G-3] `twoway_options`

## `row_options`

| `marker_options`         | change look of markers (color, size, etc.)                                  |
| `marker_label_options`   | change look or position of marker labels                                    |
| nolabel                  | remove the default row (variable) label from the graph                      |
| name(name)               | override the default name given to rows (observations)                      |
col_options | Description
---|---
\textit{pcarrow\_options} | affect the rendition of paired-coordinate arrows
\textit{nolabel} | remove the default column (variable) label from the graph
\textit{name(name)} | override the default name given to columns (variables)

See \cite{G-2} \texttt{graph twoway parrow}.

### Options

\texttt{rowover(varlist)} distinguishes groups among observations (rows) by highlighting observations on the plot for each group identified by equal values of the variables in \textit{varlist}. By default, the graph contains a legend that consists of group names. \texttt{rowover()} may not be combined with \texttt{separate} or \texttt{norow}.

\texttt{dim(# #)} identifies the dimensions to be displayed. For instance, \texttt{dim(3 2)} plots the third dimension (vertically) versus the second dimension (horizontally). The dimension numbers cannot exceed the number of variables. The default is \texttt{dim(2 1)}.

\texttt{std} produces a biplot of the standardized variables instead of the centered variables.

\texttt{alpha(#)} specifies that the variables be scaled by \(\lambda^#\) and the observations by \(\lambda^{(1-#)}\), where \(\lambda\) are the singular values. It is required that \(0 \leq # \leq 1\). The most common values are 0, 0.5, and 1. The default is \texttt{alpha(0.5)} and is known as the symmetrically scaled biplot or symmetric factorization biplot. The result with \texttt{alpha(1)} is the principal-component biplot, also called the row-preserving metric (RPM) biplot. The biplot with \texttt{alpha(0)} is referred to as the column-preserving metric (CPM) biplot.

\texttt{stretch(#)} causes the length of the arrows to be multiplied by \#. For example, \texttt{stretch(1)} would leave the arrows the same length, \texttt{stretch(2)} would double their length, and \texttt{stretch(0.5)} would halve their length.

\texttt{mahalanobis} implies \texttt{alpha(0)} and scales the positioning of points (observations) by \(\sqrt{n-1}\) and positioning of arrows (variables) by \(1/\sqrt{n-1}\). This additional scaling causes the distances between observations to change from being approximately proportional to the Mahalanobis distance to instead being approximately equal to the Mahalanobis distance. Also, the inner products between variables approximate their covariance.

\texttt{xnegate} specifies that dimension-1 \((x\) axis) values be negated (multiplied by \(-1\)).

\texttt{ynegate} specifies that dimension-2 \((y\) axis) values be negated (multiplied by \(-1\)).

\texttt{autoaspect} specifies that the aspect ratio be automatically adjusted based on the range of the data to be plotted. This option can make some biplots more readable. By default, \texttt{biplot} uses an aspect ratio of one, producing a square plot. Some biplots will have little variation in the \(y\)-axis direction, and using the \texttt{autoaspect} option will better fill the available graph space while preserving the equivalence of distance in the \(x\) and \(y\) axes.

As an alternative to \texttt{autoaspect}, the \texttt{twoway\_option aspectratio()} can be used to override the default aspect ratio. \texttt{biplot} accepts the \texttt{aspectratio()} option as a suggestion only and will override it when necessary to produce plots with balanced axes; that is, distance on the \(x\) axis equals distance on the \(y\) axis.

\texttt{twoway\_options}, such as \texttt{xlabel()}, \texttt{xscale()}, \texttt{ylabel()}, and \texttt{yscale()}, should be used with caution. These \texttt{axis\_options} are accepted but may have unintended side effects on the aspect ratio. See \cite{G-3} \texttt{twoway\_options}.
separate produces separate plots for the row and column categories. The default is to overlay the plots. separate may not be combined with rowover().

nograph suppresses displaying the graph.

table displays a table with the biplot coordinates.

---

Rows

rowopts(row_options) affects the rendition of the points plotting the rows (observations). This option may not be combined with rowover(). The following row_options are allowed:

- marker_options affect the rendition of markers drawn at the plotted points, including their shape, size, color, and outline; see [G-3] marker_options.

- marker_label_options specify the properties of marker labels; see [G-3] marker_label_options.

- mlabel() in rowopts() may not be combined with the rowlabel() option.

- nolabel removes the default row label from the graph.

- name(name) overrides the default name given to rows.

row#opts(row_options) affects rendition of the points plotting the rows (observations) in the #th group identified by equal values of the variables in varlist defined in rowover(). This option requires specifying rowover(). See rowopts() above for the allowed row_options, except mlabel() is not allowed with row#opts().

rowlabel(varname) specifies label variable for rows (observations).

norow suppresses plotting of row points. This option may not be combined with rowover().

generate(newvar_x newvar_y) stores biplot coordinates for rows in variables newvar_x and newvar_y.

---

Columns

collopts(col_options) affects the rendition of the arrows and points plotting the columns (variables). The following col_options are allowed:

- pcarrow_options affect the rendition of paired-coordinate arrows; see [G-2] graph twoway pcarrow.

- nolabel removes the default column label from the graph.

- name(name) overrides the default name given to columns.

- negcol includes negative column (variable) arrows on the plot.

- negcolopts(col_options) affects the rendition of the arrows and points plotting the negative columns (variables). The col_options allowed are given above.

- nocolumn suppresses plotting of column arrows.

---

Y axis, X axis, Titles, Legend, Overall

twoway_options are any of the options documented in [G-3] twoway_options, excluding by(). These include options for titling the graph (see [G-3] title_options) and for saving the graph to disk (see [G-3] saving_option). See autoaspect above for a warning against using options such as xlabel(), xscale(), ylabel(), and yscale().
Remarks and examples


Let $X$ be the centered (or standardized if the std option is specified) data. A biplot splits the information in $X$ into a portion related to the observations (rows of $X$) and a portion related to the variables (columns of $X$)

$$X \approx (U_2 \Lambda_2^\alpha)(V_2 \Lambda_2^{1-\alpha})'$$

where $0 \leq \alpha \leq 1$; see Methods and formulas for details. $U_2 \Lambda_2^\alpha$ contains the plotting coordinates corresponding to observations (rows), and $V_2 \Lambda_2^{1-\alpha}$ contains the plotting coordinates corresponding to variables (columns). In a biplot, the row coordinates are plotted as symbols, and the column coordinates are plotted as arrows from the origin.

The commonly used values for $\alpha$ are 0, 0.5, and 1. The default is 0.5. The \texttt{alpha()} option allows you to set $\alpha$.

Biplots with an $\alpha$ of 1 are also called principal-component biplots because $U_2 \Lambda_2$ contains the principal-component scores and $V_2$ contains the principal-component coefficients. Euclidean distance between points in this kind of biplot approximates the Euclidean distance between points in the original higher-dimensional space.

Using an $\alpha$ of 0, Euclidean distances in the biplot are approximately proportional to Mahalanobis distances in the original higher-dimensional space. Also, the inner product of the arrows is approximately proportional to the covariances between the variables.

When you set $\alpha$ to 0 and specify the \texttt{mahalanobis} option, the Euclidean distances are not just approximately proportional but are approximately equal to Mahalanobis distances in the original space. Likewise, the inner products of the arrows are approximately equal (not just proportional) to the covariances between the variables. This means that the length of an arrow is approximately equal to the standard deviation of the variable it represents. Also, the cosine of the angle between two arrows is approximately equal to the correlation between the two variables.

A biplot with an $\alpha$ of 0.5 is called a symmetric factorization biplot or symmetrically scaled biplot. It often produces reasonable looking biplots where the points corresponding to observations and the arrows corresponding to variables are given equal weight. Using an $\alpha$ of 0 (or 1) causes the points (or the arrows) to be bunched tightly around the origin while the arrows (or the points) are predominant in the graph. Here many authors recommend picking a scaling factor for the arrows to bring them back into balance. The \texttt{stretch()} option allows you to do this.

Regardless of your choice of $\alpha$, the position of a point in relation to an arrow indicates whether that observation is relatively large, medium, or small for that variable. Also, although the special conditions mentioned earlier may not strictly hold for all $\alpha$, the biplot still aids in understanding the relationship between the variables, the observations, and the observations and variables jointly.
Example 1

Gordon (1999, 176) provides a simple example of a biplot based on data having five rows and three columns.

```
. input v1 v2 v3
 v1   v2   v3
 1.  60   80  -240
 2. -213  66   180
 3.  123 -186  180
 4.  -9   38  -60
 5.   39    2  -60
 6.  end

. biplot v1 v2 v3
```

Biplot of 5 observations and 3 variables

```
Explained variance by component 1  0.6283
Explained variance by component 2  0.3717
Total explained variance 1.0000
```

The first component accounts for 63% of the variance, and the second component accounts for the remaining 37%. All the variance is accounted for because, here, the 5-by-3 data matrix is only of rank 2.

Gordon actually used an $\alpha$ of 0 and performed the scaling to better match Mahalanobis distance. We do the same using the options alpha(0) and mahalanobis. (We could just use mahalanobis because it implies alpha(0).) With an $\alpha$ of 0, Gordon decided to scale the arrows by a factor of 0.01. We accomplish this with the stretch() option and add options to provide a title and subtitle in place of the default title obtained previously.

```
. biplot v1 v2 v3, alpha(0) mahalanobis stretch(.01) title(Simple biplot) subtitle(See figure 6.10 of Gordon (1999))
```

Biplot of 5 observations and 3 variables

```
Explained variance by component 1  0.6283
Explained variance by component 2  0.3717
Total explained variance 1.0000
```
The outcome is little changed between the first and second biplot except for the additional titles and the scale of the $x$ and $y$ axes.

`biplot` allows you to highlight observations belonging to different groups by using option `rowover()`. Suppose our data come from two groups defined by variable `group`, `group=1` and `group=2`.

```
. generate byte group = cond(_n<3, 1, 2)
. list
```

<table>
<thead>
<tr>
<th></th>
<th>v1</th>
<th>v2</th>
<th>v3</th>
<th>group</th>
</tr>
</thead>
<tbody>
<tr>
<td>1.</td>
<td>60</td>
<td>80</td>
<td>-240</td>
<td>1</td>
</tr>
<tr>
<td>2.</td>
<td>-213</td>
<td>66</td>
<td>180</td>
<td>1</td>
</tr>
<tr>
<td>3.</td>
<td>123</td>
<td>-186</td>
<td>180</td>
<td>2</td>
</tr>
<tr>
<td>4.</td>
<td>-9</td>
<td>38</td>
<td>-60</td>
<td>2</td>
</tr>
<tr>
<td>5.</td>
<td>39</td>
<td>2</td>
<td>-60</td>
<td>2</td>
</tr>
</tbody>
</table>

Here is the previous biplot with group-specific markers:

```
. biplot v1 v2 v3, alpha(0) mahalanobis stretch(.01) title(Simple biplot)
  > subtitle(Grouping variable group) rowover(group)
  > row1opts(name("Group 1") msymbol(O) nolabel)
  > row2opts(name("Group 2") msymbol(T) nolabel)
```

Biplot of 5 observations and 3 variables

- Explained variance by component 1 0.6283
- Explained variance by component 2 0.3717
- Total explained variance 1.0000
In the above example, groups are defined by a single variable `group` but you can specify multiple variables with `rowover()`. The rendition of group markers is controlled by options `row1opts()` and `row2opts()`. The marker labels are disabled by using the `nolabel` option.

### Example 2

Table 7.1 of Cox and Cox (2001) provides the scores of 10 Renaissance painters on four attributes using a scale from 0 to 20, as judged by Roger de Piles in the 17th century.

```stata
. use http://www.stata-press.com/data/r14/renpainters, clear
(Scores by Roger de Piles for Renaissance Painters)
. list, abbrev(12)
```

<table>
<thead>
<tr>
<th>painter</th>
<th>composition</th>
<th>drawing</th>
<th>colour</th>
<th>expression</th>
</tr>
</thead>
<tbody>
<tr>
<td>Del Sarto</td>
<td>12</td>
<td>16</td>
<td>9</td>
<td>8</td>
</tr>
<tr>
<td>Del Piombo</td>
<td>8</td>
<td>13</td>
<td>16</td>
<td>7</td>
</tr>
<tr>
<td>Da Udine</td>
<td>10</td>
<td>8</td>
<td>16</td>
<td>3</td>
</tr>
<tr>
<td>Giulio Romano</td>
<td>15</td>
<td>16</td>
<td>4</td>
<td>14</td>
</tr>
<tr>
<td>Da Vinci</td>
<td>15</td>
<td>16</td>
<td>4</td>
<td>14</td>
</tr>
<tr>
<td>Michelangelo</td>
<td>8</td>
<td>17</td>
<td>4</td>
<td>8</td>
</tr>
<tr>
<td>Fr. Penni</td>
<td>0</td>
<td>15</td>
<td>8</td>
<td>0</td>
</tr>
<tr>
<td>Perino del Vaga</td>
<td>15</td>
<td>16</td>
<td>7</td>
<td>6</td>
</tr>
<tr>
<td>Perugino</td>
<td>4</td>
<td>12</td>
<td>10</td>
<td>4</td>
</tr>
<tr>
<td>Raphael</td>
<td>17</td>
<td>18</td>
<td>12</td>
<td>18</td>
</tr>
</tbody>
</table>
. biplot composition-expression, alpha(1) stretch(10) table
> rowopts(name(Painters)) rowlabel(painter) colopts(name(Attributes))
> title(Renaissance painters)

Biplot of 10 painters and 4 attributes

- Explained variance by component 1 0.6700
- Explained variance by component 2 0.2375
- Total explained variance 0.9075

Biplot coordinates

<table>
<thead>
<tr>
<th>Painters</th>
<th>dim1</th>
<th>dim2</th>
</tr>
</thead>
<tbody>
<tr>
<td>Del Sarto</td>
<td>1.2120</td>
<td>0.0739</td>
</tr>
<tr>
<td>Del Piombo</td>
<td>-4.5003</td>
<td>5.7309</td>
</tr>
<tr>
<td>Da Udine</td>
<td>-7.2024</td>
<td>7.5745</td>
</tr>
<tr>
<td>Giulio Rom-o</td>
<td>8.4631</td>
<td>-2.5503</td>
</tr>
<tr>
<td>Da Vinci</td>
<td>8.4631</td>
<td>-2.5503</td>
</tr>
<tr>
<td>Michelangelo</td>
<td>0.1284</td>
<td>-5.9578</td>
</tr>
<tr>
<td>Fr Penni</td>
<td>-11.9449</td>
<td>-5.4510</td>
</tr>
<tr>
<td>Perino del-Vaga</td>
<td>2.2564</td>
<td>-0.9193</td>
</tr>
<tr>
<td>Perugino</td>
<td>-7.8886</td>
<td>-0.8757</td>
</tr>
<tr>
<td>Raphael</td>
<td>11.0131</td>
<td>4.9251</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Attributes</th>
<th>dim1</th>
<th>dim2</th>
</tr>
</thead>
<tbody>
<tr>
<td>composition</td>
<td>6.4025</td>
<td>3.3319</td>
</tr>
<tr>
<td>drawing</td>
<td>2.4952</td>
<td>-3.3422</td>
</tr>
<tr>
<td>colour</td>
<td>-2.4557</td>
<td>8.7294</td>
</tr>
<tr>
<td>expression</td>
<td>6.8375</td>
<td>1.2348</td>
</tr>
</tbody>
</table>

alpha(1) gave us an \( \alpha \) of 1. stretch(10) made the arrows 10 times longer. table requested that the biplot coordinate table be displayed. rowopts() and colopts() affected the rendition of the rows (observations) and columns (variables). The name() suboption provided a name to use instead of the default names “Observations” and “Variables” in the graph legend and in the biplot coordinate table. The rowlabel(painter) option requested that the variable painter be used to label the row points (observations) in both the graph and table. The title() option was used to override the default title.

The default is to produce a square graphing region. Because the \( x \) axis containing the first component has more variability than the \( y \) axis containing the second component, there are often
no observations or arrows appearing in the upper and lower regions of the graph. The `autoaspect` option sets the aspect ratio and the \( x \)-axis and \( y \)-axis scales so that more of the graph region is used while maintaining the equivalent interpretation of distance for the \( x \) and \( y \) axes.

Here is the previous biplot with the omission of the `table` option and the addition of the `autoaspect` option. We also add the `ynegate` option to invert the orientation of the data in the \( y \)-axis direction to match the orientation shown in figure 7.1 of Cox and Cox (2001). We add the `negcol` option to include column (variable) arrows pointing in the negative directions, and the rendition of these negative columns (variables) is controlled by `negcolopts()`.

```
.biplot composition-expression, autoaspect alpha(1) stretch(10) ynegate
  > rowopts(name(Painters)) rowlabel(painter) colopts(name(Attributes))
  > title(Renaissance painters) negcol negcolopts(name(-Attributes))
```

Biplot of 10 painters and 4 attributes

- Explained variance by component 1: 0.6700
- Explained variance by component 2: 0.2375
- Total explained variance: 0.9075

---

**Stored results**

`biplot` stores the following in `r()`:

**Scalars**
- `r(rho1)`: explained variance by component 1
- `r(rho2)`: explained variance by component 2
- `r(rho)`: total explained variance
- `r(alpha)`: value of `alpha()` option

**Matrices**
- `r(U)`: biplot coordinates for the observations; stored only if the row dimension does not exceed Stata’s maximum matrix size; as an alternative, use `generate()` to store biplot coordinates for the observations in variables
- `r(V)`: biplot coordinates for the variables
- `r(Vstretch)`: biplot coordinates for the variables times `stretch()` factor
Methods and formulas

Let $X$ be the centered (standardized if $\text{std}$ is specified) data with $N$ rows (observations) and $p$ columns (variables). A biplot splits the information in $X$ into a portion related to the observations (rows of $X$) and a portion related to the variables (columns of $X$). This task is done using the singular value decomposition (SVD).

$$X = U \Lambda V'$$

The biplot formula is derived from this SVD by first splitting $\Lambda$, a diagonal matrix, into

$$\Lambda = \Lambda^\alpha \Lambda^{1-\alpha}$$

and then retaining the first two columns of $U$, the first two columns of $V$, and the first two rows and columns of $\Lambda$. Using the subscript 2 to denote this, the biplot formula is

$$X \approx U_2 \Lambda^\alpha_2 \Lambda^{1-\alpha}_2 V'_2$$

where $0 \leq \alpha \leq 1$. This is then written as

$$X \approx (U_2 \Lambda^\alpha_2)(V_2 \Lambda^{1-\alpha}_2)'$$

$U_2 \Lambda^\alpha_2$ contains the plotting coordinates corresponding to observations (rows) and $V_2 \Lambda^{1-\alpha}_2$ contains the plotting coordinates corresponding to variables (columns). In a biplot, the row coordinates are plotted as symbols and the column coordinates are plotted as arrows from the origin.

Let $\lambda_i$ be the $i$th diagonal of $\Lambda$. The explained variance for component 1 is

$$\rho_1 = \left( \sum_{i=1}^{p} \lambda_i^2 \right)^{-1} \lambda_1^2$$

and for component 2 is

$$\rho_2 = \left( \sum_{i=1}^{p} \lambda_i^2 \right)^{-1} \lambda_2^2$$

The total explained variance is

$$\rho = \rho_1 + \rho_2$$

Acknowledgment

Several biplot options were based on the work of Ulrich Kohler (2004) of Methoden der Empirischen Sozialforschung at Universität Potsdam, Germany, and coauthor of the Stata Press book *Data Analysis Using Stata.*

Kuno Ruben Gabriel (1929–2003) was born in Germany, raised in Israel, and studied at the London School of Economics and the Hebrew University of Jerusalem, earning a doctorate in demography in 1957. After several years on the faculty at the Hebrew University, he moved to the University of Rochester in 1975. His wide-ranging statistical interests spanned applications in meteorology, including weather-modification experiments, and medicine. Gabriel’s best-known contribution is the biplot.
References


Also see

[MV] ca — Simple correspondence analysis

[MV] mds — Multidimensional scaling for two-way data

[MV] pca — Principal component analysis